@model JQ_2012_RBC begin
	w[0] / c[0] ^ σ = α / (1 - n[0])

	c[0] ^ (-σ) = β * (R[0] - τ) / (1 - τ) * c[1] ^ (-σ)

	w[0] * n[0] + b[-1] - b[0] / R[0] + d[0] = c[0]

	(1 - θ) * z[0] * k[-1] ^ θ * n[0] ^ (-θ) = w[0] / (1 - μ[0] * (1 + 2 * κ * (d[0] - d[ss])))

	β * (c[0] / c[1]) ^ σ * (1 + 2 * κ * (d[0] - d[ss])) / (1 + 2 * κ * (d[1] - d[ss])) * (1 - δ + θ * (1 - (1 + 2 * κ * (d[1] - d[ss])) * μ[1]) * z[1] * k[0] ^ (θ - 1) * n[1] ^ (1 - θ)) + (1 + 2 * κ * (d[0] - d[ss])) * μ[0] * ξ[0] = 1

	(1 + 2 * κ * (d[0] - d[ss])) / (1 + 2 * κ * (d[1] - d[ss])) * (c[0] / c[1]) ^ σ * β * R[0] + (1 + 2 * κ * (d[0] - d[ss])) * μ[0] * ξ[0] * R[0] * (1 - τ) / (R[0] - τ) = 1

	b[0] / R[0] + k[-1] * (1 - δ) + z[0] * k[-1] ^ θ * n[0] ^ (1 - θ) - w[0] * n[0] - b[-1] - k[0] = (d[0] + κ * (d[0] - d[ss]) ^ 2)

	ξ[0] * (k[0] - b[0] * (1 - τ) / (R[0] - τ)) = z[0] * k[-1] ^ θ * n[0] ^ (1 - θ)

	log(z[0] / z̄) = A¹¹ * log(z[-1] / z̄) + A¹² * log(ξ[-1] / ξ̄) + σᶻ * ϵᶻ[x]

	log(ξ[0] / ξ̄) = log(z[-1] / z̄) * A²¹ + log(ξ[-1] / ξ̄) * A²² + σˣⁱ * ϵˣⁱ[x]

	y[0] = z[0] * k[-1] ^ θ * n[0] ^ (1 - θ)

	k[0] = k[-1] * (1 - δ) + i[0]

	v[0] = d[0] + c[0] * β / c[1] * v[1]

	1 + r[0] = (R[0] - τ) / (1 - τ)
end

@parameters JQ_2012_RBC begin
	b[ss] / (y[ss] * (1 + r[ss])) = BY_ratio | ξ̄
	
	n[ss] = n̄ | α

	BY_ratio = 3.36
    
    n̄ = .3

	z̄ = 1
	
	β = 0.9825
	
	σ = 1
	
	θ = 0.36
	
	δ = 0.025
	
	τ = 0.35
	
	κ = 0.146
	
	A¹¹ = 0.9457
	
	A¹² = -0.0091
	
	A²¹ = 0.0321
	
	A²² = 0.9703
	
	σᶻ = 0.0045

	σˣⁱ = 0.0098
end